

# Appendix: Mathiness in the Theory of Economic Growth by Paul Romer

This appendix exists as both a *Mathematica* notebook called Mathiness Appendix.nb and as two different pdf print-outs. The notebook is a combination of statements intended to communicate with people and statements that communicate with the computational engine in *Mathematica*.

One of the pdf printouts is called Mathiness Appendix.pdf. It prints only the plots and the typeset text that are intended as communication with people. The second pdf, called Mathiness Appendix Expanded.pdf prints those materials plus the supporting input to and detailed output from *Mathematica* that generates the plots and checks some of the algebra.

If you are reading one of these and would like any other, all three are both available for download from my website, [paulromer.net](http://paulromer.net).

## Appendix A: Scale Effects

---

As a function of individual consumption  $q$ , the individual inverse demand function is

$$p = q^{-a},$$

so as a function of total quantity  $Q$ , the inverse demand function in a market with  $N$  identical individuals gives the demand price  $p_D$  as this function of  $Q$  and  $N$ :

$$p_D = \left(\frac{Q}{N}\right)^{-a}.$$

The market inverse supply function is

$$p_S = Q^b.$$

Implicitly, these expressions assume units for measuring both output  $Q$  and the numeraire good that set any multiplicative constant in these demand functions equal to 1.

The market clearing condition with a mark-up  $m \geq 1$  that could arise from a tax or a monopoly markup is

$$p_D = m p_S,$$


which implies

$$\left(\frac{Q}{N}\right)^{-a} = m Q^b.$$

This yields a solution for  $Q$  of the form

$$Q = m^{-\frac{1}{a+b}} N^{\frac{a}{a+b}}.$$

This figure illustrates an equilibrium with a mark-up  $m = 2$ ,  $a = 0.5$ ,  $b = 1.5$ , yields a value of  $Q$  roughly equal to 4:



The shaded area represents the surplus generated in this case. To calculate the value that this area represents, we start with the area under

the individual demand curve from 0 to  $q$ , which takes the form

$$u(q) = \frac{1}{1-a} q^{1-a}.$$

As a function of the total quantity consumed, the sum across individuals of the area under the demand curves is

$$U(Q) = \frac{N}{1-a} \left( \frac{Q}{N} \right)^{1-a}.$$

Total cost, the area under the supply curve, is

$$C(Q) = \frac{1}{1+b} Q^{1+b}.$$

The surplus, equal to the shaded area in the figure, then can be calculated as

$$S = U(Q) - C(Q) = \frac{N}{1-a} \left( \frac{Q}{N} \right)^{1-a} - \frac{1}{1+b} Q^{1+b}.$$

The *Mathematica* calculations demonstrate that the implied expression for the surplus  $S$  as a function of the underlying parameters  $a$ ,  $b$ ,  $m$ , and  $N$  is

$$S = \left( \frac{(a + m - 1 + b m) m^{-\frac{1+b}{a+b}}}{(1-a)(1+b)} \right) N^{\frac{a(1+b)}{a+b}}.$$

As noted in the main text, at  $a = \frac{1}{2}$  and  $b = 0$  this reduces to

$$S = \frac{2m-1}{m^2} N.$$

## ■ Calculations

To use *Mathematica* to check the algebra in the expression offered for  $S$ , the strategy followed here is to start with the equation for  $Q$ , to transform it via a series of steps into an equivalent equation that can be trivially be solved to yield a rule that will replace the symbol  $Q$  with an expression that depends on  $m$ ,  $N$ ,  $a$ , and  $b$ . Then this rule can be applied to the

expression for  $S$  as a function of  $Q$ .

```
inverseDemand[Q_, N_] := (Q/N)^-a
(* inverse market demand *);
inverseSupply[Q_] := Q^b (* inverse market supply *);
```

```
eq1 =
  (inverseDemand[Q, N] == m * inverseSupply[Q])
(* The basic equation that determines Q;
m is the mark-up *)
```

$$\left(\frac{Q}{N}\right)^{-a} == m Q^b$$

```
e2 = PowerExpand[(m^-1 * #) & /@ eq1]
```

$$\frac{N^a Q^{-a}}{m} == Q^b$$

```
e3 = PowerExpand[(Q^a * #) & /@ e2]
```

$$\frac{N^a}{m} == Q^{a+b}$$

```
e4 = PowerExpand[(#^1/(a+b)) & /@ e3]
```

$$m^{-\frac{1}{a+b}} N^{\frac{a}{a+b}} == Q$$

```
ruleQ = Flatten[Solve[e4, Q]]
```

$$\left\{Q \rightarrow m^{-\frac{1}{a+b}} N^{\frac{a}{a+b}}\right\}$$

To calculate the surplus, we need expressions for the area under the demand and supply curves.

$$\text{areaIndividual}[q_] := \frac{1}{1-a} q^{1-a};$$

```

areaMarket[Q_, N_] := N * areaIndividual[Q / N] ;
(* sum across agents of area under individual
demand curve which equals area under market
demand curve *)

```

$$\mathbf{cost}[Q\_ ] := \frac{1}{1 + b} Q^{1+b} ;$$

```

(* cost of production equals integral under the
supply curve from 0 to Q *)

```

Now we can define surplus, apply the rule for Q, then rules that set values for a and b:

```

surplus[Q_, N_] := areaMarket[Q, N] - cost[Q] ;

```

```

s1 = FullSimplify[PowerExpand[surplus[Q, N] /. ruleQ]]

```

$$\frac{m^{-\frac{1+b}{a+b}} (-1 + a + m + b m) N^{\frac{a(1+b)}{a+b}}}{(-1 + a) (1 + b)}$$

```

s1 /. {a → 1 / 2, b → 0}

```

$$\frac{2 \left( -\frac{1}{2} + m \right) N}{m^2}$$

These calculations confirm that the expression for the surplus S as a function of the underlying parameters  $a$ ,  $b$ ,  $m$ , and  $N$  is indeed

$$S = \left( \frac{(a + m - 1 + b m) m^{-\frac{1+b}{a+b}}}{(1 - a) (1 + b)} \right) N^{\frac{a(1+b)}{a+b}},$$

which is a rearranged version of line Out[12].

They also show that  $a = \frac{1}{2}$  and  $b = 0$ , this expression reduces to

$$S = \frac{2m - 1}{m^2} N,$$

which is a rearranged version of line Out[13].

## ■ Generating the Figure that Illustrates the Surplus

Next, create a figure to illustrate the equilibrium for an example with a markup  $m = 2$  and values  $a = 0.5$ ,  $b = 1.5$ , and  $N = 1000$ .

```
inverseDemandForPlot[Q_] :=
  inverseDemand[Q, N] /. {N → 1000, a → 0.5};
inverseSupplyForPlot[Q_] := inverseSupply[Q] /. b → 1.5;
```

```
q = Q /. ruleQ
```

$$m \frac{1}{a+b} \frac{a}{N a+b}$$

```
q = q /. {N → 1000, a → 0.5, b → 1.5, m → 2}
```

```
3.97635
```

```
pS = inverseSupplyForPlot[q] (* supply price *)
```


```
7.92917
```

```
pD = inverseDemandForPlot[q]
```

```
(* demand price = m times supply price *)
```

```
15.8583
```

```
Show[
  {Plot[{inverseDemandForPlot[x],
        inverseSupplyForPlot[x]}, {x, 0, 9},
        PlotRange -> {{0, 7}, {0, 80}}],
  Plot[{inverseDemandForPlot[x],
        inverseSupplyForPlot[x], pD, pS}, {x, 0, q},
        PlotRange -> {{0, 7}, {0, 80}}, Filling -> {1 -> {2}},
        PlotLegends ->
          Placed[{"Demand", "Supply", "Demand price",
                "Supply price"}, Below],
        PlotRange -> {{0, 7}, {0, 80}},
        PlotStyle -> {Automatic, Automatic, Black,
                      {Black, Dashed}}]}]
```



Quit

---

## Appendix B: Growth in the Lucas-Moll Economies

---

To justify such statements as “the frequency at which innovations arrive,  $\beta$ , does not affect the growth rate,  $g$ ” (Lucas and Moll 2014, p. 29), the authors rely on a limit argument about a limit does not exist.

If we let  $g_\beta(t)$  denote the growth rate for a given value of  $\beta$  and date  $t$ , the rate  $g$  referenced in this quote is a limit,

$$g = \lim_{t \rightarrow \infty} g_\beta(t) = 2\%,$$

which indeed, does not depend on  $\beta$ . However, this number is not a good guide to the behavior specified in the model at any date  $T$ . The proof, available from my website as a separate comment on Lucas and Moll (2014), shows that

$$\lim_{\beta \rightarrow 0} g_\beta(T) = 0.$$

In fact, it establishes a stronger result, that the growth rate converges uniformly to zero on any finite interval:

**Proposition:** *In the family of B economies, for any  $T > 0$  and any  $\epsilon > 0$ , there exists a  $\tilde{\beta} > 0$  such that for all  $\beta \in (0, \tilde{\beta})$  and all  $t \in [0, T]$ ,*

$$0 \leq g_\beta(t) < \epsilon.$$

So at every date  $T$ , the growth rate does depend on  $\beta$  and it converges to zero as  $\beta$  goes to zero. This is exactly what one should expect from the model itself. When  $\beta$  is set equal to 0, the model implies that the growth rate is also equal to zero.

The difference between these two types of limit implies that the order one uses to calculate the double limit matters:

$$\lim_{T \rightarrow \infty} \lim_{\beta \rightarrow 0} g_\beta(T) \neq \lim_{\beta \rightarrow 0} \lim_{T \rightarrow \infty} g_\beta(T).$$


This dependence on the order tells us that if we recognize  $g(\beta, T)$  as a function from  $\mathbb{R}^2 \rightarrow \mathbb{R}$ , the limit of  $g$  as  $(\beta, T)$  goes to  $(0, \infty)$  does not exist.

Lucas and Moll use limit arguments about this limit that does not exist to claim that two types of economy, labeled here the P and B economies,




are so similar as to be observationally equivalent. Simple numerical calculations show the growth rate in these two types of economy differ in ways that are plainly observable.

For example, this figure shows a plot over time of the growth rate in a version of the P economy.



Contrast this with the next figure, which shows the plot over time of the growth rate in a collection of B economies with decreasing arrival rates  $\beta$ , where curves displaced to the right have a lower value of  $\beta$ . It shows how the growth rate at any fixed date such as  $T = 300$  falls from a value close to 2% to a value close to zero as  $\beta$  decreases.



These plots show that there at every date  $T$  is no value for the growth rate  $g$  that the P economy shares with the collection of all the B economies. The only value that they have in common is the limit of the growth rate for fixed  $\beta$  as  $T$  goes to infinity. As noted in the text of the accompanying article, this kind of limit is not an observable. No feasible set of observations on the growth rate, which must of course be taken at finite times, can ever falsify an assertion about a limit as time goes to infinity.

## ■ Plotting Growth in the P Economy

To calculate the growth rate in the P or Pareto economy, start by expressing the equation for  $F$  as input to *Mathematica*:

$$F[z_, t_] := \left( e^{(\alpha + \beta (1 - G[z])) * t} * \left( \frac{1}{H[z]} - \frac{\alpha}{\alpha + \beta (1 - G[z])} \right) + \frac{\alpha}{\alpha + \beta (1 - G[z])} \right)^{-1}$$

To have a formal expression that corresponds closely to an expression like  $F(z, t)$  that we use in typeset math, this definition makes  $z$  and  $t$  the only explicit arguments for the function  $F$ . But as is often the case, the function also depends on a set of parameters that can also change. In particular, in the next section, we will want to consider a sequence of values for  $\beta$ . This means that any time we invoke  $F$ , it will be important

to keep track of whether a particular value of  $\beta$  has been set in the background or whether instead  $\beta$  remains an unspecified placeholder.

As noted in the proof above, if we can use a change of variable  $x = z - 1$  to specify the distributions  $H$  and  $G$ , the expression for  $F_\beta$  as a function of  $x$  is unchanged.

With this change of variable in mind, we can specify the expression for  $H$  and some of the parameters, define the special case of  $F$  for the P economy, which we denote by  $FP$ , and then calculate its PDF,  $fP$ :

```
H[x_] := Refine[CDF[FrechetDistribution[ $\psi$ , s], x], x > 0];
H[0] = 0;
 $\psi$  = 2; (* In Mathematica's parameterization,
the tail parameter  $\psi$  is the inverse of the
parameter  $\theta$  used by Lucas and Moll,
so  $\theta^{-1} = \psi = 2$  implies  $\theta = 0.5$  *)
s = 1;
 $\beta$  = 0;
```

```
FP[x_, t_] = F[x, t]
```

$$\frac{1}{1 + e^{t\alpha} \left( -1 + e^{\frac{1}{x^2}} \right)}$$

```
fP[x_, t_] = D[FP[x, t], x]
```

```
(* The probability density function associated
with FP(x,t) . *)
```

$$\frac{2 e^{\frac{1}{x^2} + t\alpha}}{\left( 1 + e^{t\alpha} \left( -1 + e^{\frac{1}{x^2}} \right) \right)^2} x^3$$

Next, for this P Economy, we calculate a table of values of output at 10 year intervals over the range  $t = 0$  to 500 and take the difference of the logarithms divided by 10 as the growth rate for an interval of time centered at the midpoint of the 10 year interval.

Because  $FP(\cdot, t)$  is absolutely continuous (i.e. is equal to the integral of its derivative), we can calculate the mean productivity in the economy, which is equal to output, by taking an integral with respect to the pdf

fP( $\cdot$ ,  $t$ ):

$$y_{\beta}(t) = \int_0^{\infty} (x + 1) fP(x, t) dx.$$

```
α = 0.04 (* specify the last remaining parameter. *)
0.04
```

```
outputPEconomy =
```

```
Table[NIntegrate[(x + 1) * fP[x, t], {x, 0, ∞},  
AccuracyGoal → Automatic], {t, 0, 500, 10} ]
```

```
{2.77245, 3.09981, 3.50458, 4.00421, 4.61986,  
5.37718, 6.3074, 7.44851, 8.84685, 10.559, 12.6539,  
15.2159, 18.348, 22.1761, 26.8538, 32.569, 39.551,  
48.0802, 58.4988, 71.2251, 86.7696, 105.756,  
128.947, 157.274, 191.871, 234.129, 285.744, 348.786,  
425.786, 519.835, 634.705, 775.009, 946.377, 1155.69,  
1411.34, 1723.59, 2104.97, 2570.8, 3139.76, 3834.69,  
4683.48, 5720.19, 6986.44, 8533.03, 10422., 12729.3,  
15547.4, 18989.4, 23193.5, 28328.3, 34600.1}
```

```
growthPEconomy =
```


```
(Log[Take[outputPEconomy, -50]] -  
Log[Take[outputPEconomy, 50]]) / 10;
```

```
plotDataGrowthPEconomy =
```

```
Table[{10 * j - 5, growthPEconomy[[j]]}, {j, 1, 50}];
```

The figure that illustrates the implied growth rates shows that in this example, the growth rate for the P economy converges to its steady state value of 2 % per year in roughly 200 years.

```
ListPlot[plotDataGrowthPEconomy,
  PlotRange → {{0, 500}, {0, .025}}, Joined → True,
  PlotLabel → "P Economy",
  AxesLabel → {"Year", "Growth Rate"},
  AxesStyle → AbsoluteThickness[1]]
```



## ■ Plotting Growth in the B Economies

For the B economies considered here, the initial distribution  $H(z)$  has all its mass at  $z = 1$  and receives exogenous idea injections from a Frechet distribution defined on the interval  $[1, \infty)$ . As in the proof, it simplifies the analysis to use the change of variable  $x = z - 1$  so the initial distribution of productivity  $H(x)$  has its mass of 1 at  $x = 0$ .

For the new calculations, we must clear the expressions for  $H$  and  $\beta$ , then define the new expressions for  $H(x)$  and  $G(x)$ . It is also convenient to clear the definition of  $\alpha$  so we can calculate an expression for  $F_\beta(0, t)$  that shows the role of  $\alpha$  as a placeholder:

```
ClearAll[ $\beta$ ,  $\alpha$ , H];
H[x_] = 1 (* valid for  $x \in [0, \infty)$  *);
G[x_] := Refine[CDF[FrechetDistribution[ $\psi$ , s], x],
  x > 0];
G[0] = 0;
```

Given the assumptions about  $H$ , and  $G$ ,  $F$  takes the form:

$$\mathbf{FB}[\mathbf{x}_-, \mathbf{t}_-] = \mathbf{F}[\mathbf{x}, \mathbf{t}]$$

$$1 / \left( \frac{\alpha}{\alpha + \left(1 - e^{-\frac{1}{x^2}}\right) \beta} + e^{\mathbf{t} \left( \alpha + \left(1 - e^{-\frac{1}{x^2}}\right) \beta \right)} \left( 1 - \frac{\alpha}{\alpha + \left(1 - e^{-\frac{1}{x^2}}\right) \beta} \right) \right)$$

Because  $H$  has a mass point at 0, so does  $F_\beta(\cdot, t)$ :

$$F_\beta(0, t) = \frac{\alpha + \beta}{\alpha + \beta e^{(\alpha + \beta)t}} > 0$$

To confirm this expression:


$$\mathbf{FB}[\mathbf{0}, \mathbf{t}_-] = \mathbf{Limit}[\mathbf{FB}[\mathbf{x}, \mathbf{t}], \mathbf{x} \rightarrow \mathbf{0}]$$

$$\frac{\alpha + \beta}{\alpha + e^{\mathbf{t}(\alpha + \beta)} \beta}$$

With probability  $\beta$ , workers get a draw from the distribution  $G(x)$ , which has the following density:

$$\mathbf{g}[\mathbf{x}_-] = \mathbf{D}[\mathbf{G}[\mathbf{x}], \mathbf{x}];$$

```
Plot[g[x], {x, 0.01, 5}, PlotRange -> {{0, 5}, {0, 1}},
  AxesStyle -> AbsoluteThickness[1]]
```



To calculate output, we need both the mass point  $F_\beta(0, t)$  and the function  $f_\beta(x, t)$  that is the partial derivative of  $F_\beta(x, t)$  with respect to  $x$ . Because of the change of variable, output is the average value of  $x + 1$ ,

$$y_\beta(t) = \int_0^\infty (x + 1) dF_\beta(x, t),$$

which in this case must be interpreted as a Riemann-Stieltjes integral, and can be expressed as

$$y_\beta(t) = F_\beta(0, t) + \int_0^\infty (x + 1) f_\beta(x, t) dx.$$

```
ClearAll[β];
fB[x_, t_] = D[FB[x, t], x]
(* Clear β so that the expression for fB has β
  as a placeholder. This way we can change the
  value for β without having to redefine fB. *)
β = 2-12;
α = 0.04;
```

As one final consistency check, we can verify that  $F_\beta(0, t)$  and

$\int_0^\infty f_\beta(x, t) dx$  add up to 1 at each time  $t$ , as they should.


```

plotDataForf =
  Table[{t, NIntegrate[fB[x, t], {x, 0, ∞},
    AccuracyGoal → Automatic]}, {t, 0, 500, 10}];

plotDataForF0 = Table[{t, FB[0, t]}, {t, 0, 500, 10}];

ListPlot[{plotDataForf, plotDataForF0}, Joined → True,
  PlotLegends → Placed[{"Prob( x > 0 )", "Prob( x = 0 )"},
    Below], AxesStyle → AbsoluteThickness[1]]

```



To plot growth rates for the B economies, it helps to define a module that takes a value of  $n$  as input and sets  $\beta = 2^{-n}$ . With this value, it calculates output every 10 years in the range 0 to 500. The growth rate centered at the midpoint of any interval is the difference of the log of output at two dates divided by 10.

```
ClearAll[ $\beta$ ];
```



```

growthBEconomy[n_] := Module[
  {outputBEconomy1, outputBEconomy2},
   $\beta = 2^{-n}$ ;
  outputBEconomy1 = Table[N[FB[0, t]], {t, 0, 500, 10}];
  outputBEconomy2 =
    Table[NIntegrate[(x + 1) * fb[x, t], {x, 0,  $\infty$ },
      AccuracyGoal -> Automatic], {t, 0, 500, 10}];
  (Log[Take[outputBEconomy1 + outputBEconomy2, -50]] -
    Log[Take[outputBEconomy1 + outputBEconomy2, 50]]) /
    10
]

```

Warning: On a laptop running *Mathematica*, evaluating the next cell can take on the order of 10 minutes.


```

growthB12 = Table[{10 * i - 5, growthBEconomy[12][[i]]},
  {i, 1, 50}];
growthB16 = Table[{10 * i - 5, growthBEconomy[16][[i]]},
  {i, 1, 50}];
growthB20 = Table[{10 * i - 5, growthBEconomy[20][[i]]},
  {i, 1, 50}];
growthB24 = Table[{10 * i - 5, growthBEconomy[24][[i]]},
  {i, 1, 50}];
growthB28 = Table[{10 * i - 5, growthBEconomy[28][[i]]},
  {i, 1, 50}];
growthB32 = Table[{10 * i - 5, growthBEconomy[32][[i]]},
  {i, 1, 50}];
growthB36 = Table[{10 * i - 5, growthBEconomy[36][[i]]},
  {i, 1, 50}];

```

■ Results

```
ListPlot[{growthB12, growthB16, growthB20, growthB24,
  growthB28, growthB32, growthB36},
  PlotRange -> {{0, 500}, {0, .025}}, Joined -> True,
  GridLines -> Automatic,
  AxesLabel -> {"Year", "Growth Rate"},
  PlotLabel ->
    "B Economy;  $\beta$  is smaller for curves displaced
    to the right", AxesStyle -> AbsoluteThickness[1]]
```



Quit

## Appendix C: Calculating the Ratio of Wealth to Income

---

Let  $Y_\gamma$  denote income measured in gross terms and let  $Y_\nu$  denote income measured in net terms. (The mnemonic is “gamma for gross” and “nu for net.”) Symmetrically, let  $s_\gamma$  and  $s_\nu$  denote the gross and net measures of saving. Let  $\delta$  be the depreciation rate. Let  $g$  be the rate of growth of gross output. Let  $K$  be the stock of capital.

By definition,

$$Y_\nu = Y_\gamma - \delta K,$$

$$\dot{K} = s_\gamma Y_\gamma - \delta K = s_\nu Y_\nu.$$

For  $\frac{K}{Y_\gamma}$  to be constant in a steady state,  $\frac{\dot{K}}{K}$  must be equal to  $g$ , which implies

$$g = \frac{\dot{K}}{K} = s_\gamma \frac{Y_\gamma}{K} - \delta = s_\nu \frac{Y_\nu}{K}.$$

This yields the two expressions for the steady-state ratio of capital to output,

$$\frac{K}{Y_\gamma} = \frac{s_\gamma}{g + \delta}, \quad (1)$$

$$\frac{K}{Y_\nu} = \frac{s_\nu}{g}. \quad (2)$$

From the definitions, the ratio of  $Y_\gamma$  to  $Y_\nu$  is equal to

$$\frac{Y_\gamma}{Y_\nu} = \frac{Y_\nu + \delta K}{Y_\nu} = 1 + \delta \frac{K}{Y_\nu} = 1 + \delta \frac{s_\nu}{g} \quad (3)$$

By taking the ratio of equation (2) to equation (1), we have a second expression for the ratio of gross to net income

$$\frac{Y_\gamma}{Y_\nu} = \frac{s_\nu}{s_\gamma} \frac{g + \delta}{g}. \quad (4)$$

Combining this with equation (3) yields

$$1 + \delta \frac{s_\gamma}{g} = \frac{s_\gamma}{s_\gamma} \frac{g + \delta}{g}.$$

Solving for  $s_\gamma$  yields

$$s_\gamma = s_\gamma \frac{g + \delta}{g} \frac{1}{1 + \delta \frac{s_\gamma}{g}} = s_\gamma \frac{g + \delta}{g} \frac{g}{g + \delta s_\gamma} = s_\gamma \frac{g + \delta}{g + \delta s_\gamma}. \quad (5)$$

Let  $g$  be an initial growth rate and  $g^*$  be a lower growth rate. In the example considered by Piketty and Zucman (2014),  $g^* = \frac{1}{2} g$ . Following their approach, assume that  $s_\gamma$  stays constant when  $g$  changes. Let other variables with an asterisk denote the values implied by the new saving rate. For example, let  $s_\gamma^*$  denote the new gross saving rate implied by  $g^*$ .

Consider the ratio

$$\frac{g}{g^*} = \frac{\left(\frac{K}{Y_\gamma}\right)^*}{\frac{K}{Y_\gamma}}.$$

To decompose this into the three terms described in the text, rewrite the numerator and denominator in terms of ratios that involve  $Y_\gamma$

$$\frac{g}{g^*} = \frac{\left(\frac{K}{Y_\gamma}\right)^* \left(\frac{Y_\gamma}{Y_\gamma}\right)^*}{\left(\frac{K}{Y_\gamma}\right) \frac{Y_\gamma}{Y_\gamma}}.$$

Then use the expression for  $\frac{K}{Y_\gamma}$  from equation (1) to rewrite the first term,

$$\frac{g}{g^*} = \frac{\frac{s_\gamma^*}{g^* + \delta} \left(\frac{Y_\gamma}{Y_\gamma}\right)^*}{\frac{s_\gamma}{g + \delta} \frac{Y_\gamma}{Y_\gamma}}$$

and multiply and divide by  $s_\gamma$  to get

$$\frac{g}{g^*} = \left(\frac{\frac{s_\gamma}{g^* + \delta}}{\frac{s_\gamma}{g + \delta}}\right) \left(\frac{s_\gamma^*}{s_\gamma}\right) \left(\frac{\left(\frac{Y_\gamma}{Y_\gamma}\right)^*}{\frac{Y_\gamma}{Y_\gamma}}\right) = T_1 T_2 T_3.$$

The first of these three terms,

$$T_1 = \frac{\frac{s_\gamma}{g^* + \delta}}{\frac{s_\gamma}{g + \delta}},$$

is the change in the ratio of capital to gross income that we would observe if the gross saving rate remained constant at its initial value  $s_\gamma$  when the growth rate falls from  $g$  to  $g^*$ . The second term,

$$T_2 = \frac{s_\gamma^*}{s_\gamma} = \frac{\frac{g^* + \delta}{g^* + \delta s_\gamma}}{\frac{g + \delta}{g + \delta s_\gamma}}$$

captures the change in the gross saving rate that is implied by the change in  $g$  when  $s_\gamma$  remains constant. The final term,

$$T_3 = \frac{\left(\frac{Y_\gamma}{Y_\nu}\right)^*}{\frac{Y_\gamma}{Y_\nu}} = \frac{1 + \delta \frac{s_\nu}{g^*}}{1 + \delta \frac{s_\nu}{g}},$$

is the change in the ratio of gross to net income implied by the change in the growth rate.

The parameter values in the paper,  $\delta = 3\%$ ,  $s_\nu = 10\%$ ,  $g = 3\%$ , and  $g^* = 1.5\%$  imply values

$$T_1 = 1.333,$$

$$T_2 = 1.375, \text{ and}$$

$$T_3 = 1.091.$$

In this example, because  $g$  falls to half its previous value, the ratio of  $K$  to  $Y_\nu$  doubles. The calculations verify that with these values, the product  $T_1 T_2 T_3$  is equal to 2.

## ■ Calculations for steady state relationships

$$\mathbf{T1} = \frac{\mathbf{g} + \delta}{\mathbf{g}^* + \delta}$$

$$\frac{\mathbf{g} + \delta}{\delta + \mathbf{g}^*}$$

$$\mathbf{T1} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_v \rightarrow 0.1\}$$

1.33333

$$s_\gamma = s_v \frac{g + \delta}{g + \delta s_v} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_v \rightarrow 0.1\}$$

0.181818

$$s_\gamma^* = s_v \frac{g^* + \delta}{g^* + \delta s_v} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_v \rightarrow 0.1\}$$

0.25

$$\mathbf{T2} = s_\gamma^* / s_\gamma$$

1.375

$$\mathbf{T3} = \frac{1 + \delta \frac{s_v}{g^*}}{1 + \delta \frac{s_v}{g}} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_v \rightarrow 0.1\}$$

1.09091

$$\mathbf{T1 * T2 * T3} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_v \rightarrow 0.1\}$$

2.

## ■ Dynamics

Because it is an easy exercise, it may also be of some interest to explore the time path for these variables. Consider a case of an economy that starts in a steady state with  $g = 3\%$  and  $s_v = 10\%$  so the implied value for  $s_\gamma$  is approximately  $18\%$ . We can calculate the capital to gross output ratio as

$$\frac{K}{Y_\gamma} = \frac{s_\gamma}{g + \delta}.$$

Let  $k = \frac{K}{Y_\gamma}$  denote this ratio and consider the expression for its percentage rate of change

$$\frac{\dot{k}}{k} = \frac{\dot{K}}{K} - \frac{\dot{Y}_\gamma}{Y_\gamma} = s_\gamma \frac{Y_\gamma}{K} - \delta - g,$$

which implies

$$\dot{k} = s_\gamma - (\delta + g)k, \quad (6)$$


where  $s_\gamma$  is related to  $s_v$  by equation (9) from above,

$$s_\gamma = s_v \frac{g + \delta}{g + \delta s_v}. \quad (7)$$

Suppose that at time zero, the economy is at its steady state at a growth rate  $g = 3\%$  per year. At time zero, the growth rate falls to from  $3\%$  to  $1.5\%$ .

In an approach that maintains that the saving rate is fixed when the growth rate changes, there are now two possibilities -- either the gross saving rate remains constant or the net saving rate remains constant. To capture the first case, we can use a version of equation (6) uses  $g^* = 1.5\%$  but leaves  $s_\gamma$  unchanged at roughly  $18\%$ . In the second case, in equation (6) we insert both the new value  $s_\gamma^* = 25\%$  implied by equation (7) and the new growth rate  $g^*$ .

This plot compares the evolution of the capital to gross output ratio under these two scenarios.



## ■ Calculations for the dynamic paths

$$s_y$$

$$0.181818$$

$$s_y^*$$

$$0.25$$

$$b = \delta + g^* /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_y \rightarrow 0.1\}$$

$$0.045$$

$$k0 = \frac{s_y}{g + \delta} /. \{g \rightarrow 0.03, g^* \rightarrow 0.015, \delta \rightarrow 0.03, s_y \rightarrow 0.1\}$$

$$3.0303$$

$$r1 = Flatten[DSolve[{k'[t] + b*k[t] - a == 0, k[0] == k0}, k[t], t]]$$

$$\{k[t] \rightarrow 22.2222 e^{-0.045 t} (0.136364 - 1. a + 1. a e^{0.045 t})\}$$

$$k[t] = k[t] /. r1$$

$$22.2222 e^{-0.045 t} (0.136364 - 1. a + 1. a e^{0.045 t})$$



## ■ Graph

```

Plot[{Evaluate[k[t] /. {a → sy*}],
      Evaluate[k[t] /. {a → sy}}]}, {t, 0, 100},
PlotRange → {{0, 100}, {0, 6}},
PlotLegends → Placed[{"sv constant", "sy constant"},
                      Below],
AxesLabel → {"Year", "Capital to Gross-Output Ratio"},
AxesStyle → AbsoluteThickness[1]]

```

